# This file is part of test-drive.
# SPDX-Identifier: Apache-2.0 OR MIT
#
# Licensed under either of Apache License, Version 2.0 or MIT license
# at your option; you may not use this file except in compliance with
# the License.
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.

project(
  'test-drive',
  'fortran',
  version: '0.4.0',
  license: 'Apache-2.0 OR MIT',
  meson_version: '>=0.60',
  default_options: [
    'buildtype=debugoptimized',
    'default_library=both',
  ],
)

# General configuration information
lib_deps = []
subdir('config')

# Collect source of the project
srcs = []
subdir('src')

# MCTC library target
testdrive_lib = library(
  meson.project_name(),
  sources: srcs,
  version: meson.project_version(),
  dependencies: lib_deps,
  install: true,
)

# Export dependency for other projects and test suite
testdrive_inc = testdrive_lib.private_dir_include()
testdrive_dep = declare_dependency(
  link_with: testdrive_lib,
  include_directories: testdrive_inc,
  dependencies: lib_deps,
)

# Package the license files
testdrive_lic = files(
  'LICENSE-Apache',
  'LICENSE-MIT',
)

install_data(
    testdrive_lic,
    install_dir : get_option('datadir')/'licenses'/meson.project_name(),
)
if host_machine.system() == 'windows'
    symbols_file = 'lib'+meson.project_name()+'-'+meson.project_version().split('.')[0]+'.dll.symbols'
    obj_suffix = '.obj'
else
    symbols_file = 'lib'+meson.project_name()+'.so.'+meson.project_version()+'.symbols'
    obj_suffix = '.o'
endif
install_subdir(testdrive_lib.full_path()+'.p',
    install_dir: 'include'/meson.project_name(),
    strip_directory: true,
    exclude_files: [
        'depscan.dd',
        meson.project_name()+'-deps.json',
        symbols_file,
        meson.project_name()+'.dat',
        'src_testdrive.F90'+obj_suffix,
        'src_testdrive_version.f90'+obj_suffix,
    ]
)

pkg = import('pkgconfig')
pkg.generate(
    testdrive_lib,
    name: meson.project_name(),
    description: 'The simple testing framework.',
    version: meson.project_version(),
    subdirs: meson.project_name(),
)

# add the testsuite
subdir('test')
